SQL Server Logon Trigger Implications - are there any side effects possible in my scenario?
Database mail uses a Service Broker queue and an asynchronous background process to actually send emails, so performance shouldn't be a big deal.  But logon triggers can easily cause downtime, so they require a great deal of care in writing and testing.  Also you could end up with thousands of emails if a developer runs a load test or somesuch.
So it's probably overkill to use a logon trigger for this.  Instead use an Audit or even just an XEvent session.  Write the data to an event file and processes it with a scheduled job.
Here's how to create and query an Audit:
--ALTER SERVER AUDIT [sa_successful_logon] WITH (STATE = OFF);
--DROP SERVER AUDIT [sa_successful_logon]
go

declare @logFolder nvarchar(255) = ( select [path]  from sys.dm_os_server_diagnostics_log_configurations )
exec('
CREATE SERVER AUDIT [sa_successful_logon]
TO  FILE ( FILEPATH = '''+@logFolder+''' )
WHERE ([server_principal_name]=''sa'');
ALTER SERVER AUDIT [sa_successful_logon] WITH (STATE = ON);
')
GO

CREATE SERVER AUDIT SPECIFICATION [sa_successful_logon_logon_spec]
FOR SERVER AUDIT [sa_successful_logon]
ADD (SUCCESSFUL_LOGIN_GROUP)
WITH (STATE = ON)

go

declare @log nvarchar(255) = ( select [path] + 'sa_successful_logon*.sqlaudit' from sys.dm_os_server_diagnostics_log_configurations ) 
SELECT event_time, client_ip, application_name
FROM sys.fn_get_audit_file (@log,default,default);